Arrays-VB.Net
1. One-dimensional Arrays

An array is a collection of data in the form of a list , but it is not visible on screen.  In VB, we can declare arrays, and give them meaningful names. We can then select individual elements by an integer subscript.  The array exists in RAM, and can be accessed very quickly (e.g. 1 microsecond per element). 
Imagine an array as a street - each house has a number 0, 1, 2, etc, but inside each house there are a number of people.  This number bears no relation to the house number.  For example:

dim house (100) as integer     ' an array, numbered 0 up to 100

house(3) = 2         ' 2 people live in house 3

house(65) = 0        ' no=one in house 65

messageBox.show("people: "  &  house(3))     '  displays 2
Round brackets must surround the subscript.

We declare arrays at the same place in the program as other variables, e.g:


dim x as integer

  dim a(3) as integer   ' see below


dim  age(70) as single


dim house(100) as integer

When we declare the array, we specify the type of each element and the highest subscript.  The first element has a zero subscript.  Thus, we have declared an array called 'a', with elements


a(0) , a(1) , a(2) , a(3)

When we declare an array, we provide its highest position number.
Pictorially, we can represent it as:

	a(0)
	

	a(1)
	

	a(2)
	

	a(3)
	


As you can see, this is a single column of 'boxes', each of which can hold a number.  For this reason, it is called a 'one-dimensional' array.  Later, we will see that a two-dimensional array is a 'grid' of boxes, like a chess board. 

You may be familiar with arrays in other languages - if so you will be aware that the above numbering system is slightly different, in that subscripts always begin at zero.  Also, in C++, Java you state the number of elements in the array.

It is not possible to perform operations on a whole array with one statement - we would have to write functions or procedures to do this - but individual elements are just like variables, and can be manipulated in the same way  -  we can write statements of the form:


a(2) = b(x-8) * (a(0) - 4)


MessageBox.show( "value is": &  2*a(x*x+3) )
Subscripts in () can be  expressions.

2. Range of Subscripts

Consider


dim a(3) as integer

The programmer intends (with the best will in the world) to only use subscripts for 'a' in the range 0 to 3.  but  the programmer could get it wrong.  For example, statements such as:


n = CInt(inputBox("type a number"))
 'read n from user


x = a(n)

will give a run-time error if '4' is typed.  The safe version of the above would have an 'if' to check if n was numeric, and was in the right range ,   and to prevent execution of the second statement in certain cases.  

Problem
a) Write a program containing an array named person.  It should be able to hold 5 strings.

Use 5 input boxes to input names into the 5 elements.

b) Now, use 5 MessageBoxes  to display all the names, one after another.

c) Use an input box to get a number (between 0 and 4) from the user.  Display the corresponding name.

Array Input/Output

When you try some array problems, you will have the incidental problem of getting data into and out of the array  - below we give some code which can be used.  It has:

· a quit button

· an enter button, which produces a series of input requests, storing the data into an array.

· A display button, which  sends the elements  to  a text box.

(the complete code is shown at the end of this chapter)
Dim a(9) As Integer    'global
Dim nextPlace As Integer

'NB the text2 box must have both scrollbars on, AND set
'   multi-line to true

' do these at design time

Private Sub DisplayButton_Click(etc...
    Text2.Text = ""      ' clear the big box

    Dim place As Integer

    For place = 0 To 9 
        Text2.AppendText(place & "  " & a(place)& vbCrLf)
    Next 
End Sub

Private Sub EnterButton_Click(etc...)

    Dim place As Integer

    For place = 0 To 9

        a(place) = CInt(InputBox("Enter item " & place))
    Next 
End Sub

private sub QuitButton_click(etc...)

    end

end sub

3.  Reversing Numbers

Problem - input 1000 numbers, and display them in reverse order!  Note that we can't start to display the numbers until we have input the last one.  We use an array to hold all the numbers.  (To test this, use e.g 10 instead of 1000 !)
Here is a solution:

'reversing a sequence of 1000 numbers Mk 1  

    dim data(1000) as integer


 private sub GoButton_Click(etc...)   'start with  this button



 dim  n as integer



 for n = 1 to 1000





     ' read number into n'th element  



     data(n) = CInt(inputBox("type next value"))


 next 
    for n = 1000 to 1 Step -1   ' go from 1000, 999, to 1

          ' add element to  a text box, with end-of-line

          text1.AppendText(data(n) & VBCrLf) 

    next
    end sub

In the above, we could have declared the array locally to the Go Button sub, but in many problems, the array will be declared at a global form level, to enable - for example - one sub to input the data, and other subs to process it.

We could also pass the array as a parameter to procedures.

We decided not to use element (0), preferring to count from 1 to 1000, rather than 0 to 999.  This is a clarity issue - most humans count from 1 upwards.

Many array problems involve the above pattern of doing the same operation on each element, thus the above 'for' loop is very common.  Only the part between the 'for'  and    'next'  will change.  Assuming that data has been read into the array, here is how we could add up all the elements:

  dim n as integer

  dim sum as integer


sum = 0


for n = 1 to 1000 


   
sum = sum + data(n)


next
Returning to our house analogy, we could use the above pattern to add up the m=number of people in the whole street.  If the houses had names, not numbers, we could not use a loop and a subscript.
4. Searching

Frequently, we need to search an array for a particular value.  When the contents of the array are in a particular order ( e.g. ascending ), we might consider an algorithm which looks at the middle element, and then decides which half of the data to examine - rather in the way one might search a dictionary.  However,  we will assume that the values in the array have no pattern to them, and we will step through them one at a time, stopping when we find the required one.  This is analogous to the compact disc search in the earlier design chapter.  Note that, when searching there are two conditions:  we keep looking as long as we haven't found the value, and as long as we haven't reached the end of the array.

Here is the complete program - because we have not yet covered arrays as arguments , it has not been split into functions.  The program reads in 20 numbers, then asks the user to type a number.  It is reported as not found, or found at a particular position. 


' search array sequentially 

   private sub Go_ButtonClick(etc...)


dim data(20) as integer


dim place, wanted as integer

   dim found as boolean


' read in the data  


for place = 1 to 20



   
data(place) = CInt(inputBox("type data value"))

next place


' request the data value to search for 


wanted = CInt(inputBox("which number to look for?"))

  ' do the search 


found = false

' false means  'not yet' 


place = 1


while  (place <=20) and ( found = false )


    
if  data(place)= wanted then




     
found = true


    
else



       

place = place + 1

        end if


wend

     ' display result


if found = false then


    
MessageBox.show( wanted  &  " was not found " )

else


    
MessageBox.show( wanted  & " found at position " & place)
  end if

end sub

Note the output is of the form:



235 found  at position 13

In most search applications, it is vital that we know the position of the result - we may use it modify the value, or as a subscript for a second array.

5 Finding the Largest

Our task is to examine the array elements to find the one containing the largest.  Again, it is important to find the position of the largest.

The algorithm assumes that the first one is the largest, then examines the rest of the elements in turn.  A 'largest place so far' variable is used, which is reset if need be.  Below, we sketch out the main bits:


' largest in an array 



dim  largePlace as integer



' read 20 values into data(), as previous program 



' find largest 



largePlace = 1

' assume element 1 is largest 



for place = 2 to 20 
'yes, 2 !



if data(place)>data(largePlace) then





   largePlace = place

      end if



next 


MessageBox.show(("largest:" & data(largePlace) & ", position:" & _   

              largePlace))

end sub

The output is of the form:


largest: 2365 ,  position: 13

Note that we used   space   followed by   _   to enter a long line split in two.

6. Arrays As Parameters

In VB we can pass arrays to procedures and functions, and can make use of the built-in   Ubound function to find the allowed range of subscripts.  So ,  we can write procedures which can work with arrays of any size.  Here is a program showing a function which adds up and returns the elements of ANY int array.

Private Function sumArray(byVal a() As Integer) As Integer                 
    Dim total as Integer = 0

    Dim n As Integer

    For n = 0 To UBound(a)

        total = total + a(n)
    Next
    Return total 
End Function
'now try calling it:

Private Sub GoButton_Click()

    Dim x(6) As Integer

    Dim b(2) As Integer
    Dim m As Integer
    'fill x and b with data  - omitted here

    m = sumArray(x)
    MessageBox.Show("Sum of x is: " & m)

    m = sumArray(b)

    MessageBox.Show("Sum of b is: " & m)
End Sub

Though arguments (sometimes called parameters) are initially a difficult topic, they are difficult in all languages and are widely used by all professional programmers..

7.  2-Dimensional Arrays - Introduction

Previously we have looked at one-dimensional arrays, which can be imagined as a single column (or row, depending on how we choose to draw them) of items.  Here we advance to two-dimensional arrays, which can be imagined as a 'grid' of variables.  Each element now needs two subscripts to identify it..

You can think of it rather like a spreadsheet - the main difference in concept is that spreadsheet is visible, whereas an array in held in RAM  - if we need to see its contents, we must write code - e.g. nested loops - to display it.

In fact, if we want some form of 2-dim display, we might go for an add-in VB control, such as the 'grid'.  Arrays are used more for  internal storage.

8.  Declaring and Accessing

Arrays in VB can have up to 60 dimensions, but most programs only go up as far as 2.  Although computers know nothing about the concepts of horizontal and vertical, it is useful to imagine such arrays as:










cols....

	
	    0
	   1
	   2
	   3

	   0
	
	
	
	

	   1
	
	
	
	

	   2
	
	
	
	



     rows

There are 4 columns and 3 rows - recall that we can use subscripts from zero in VB, and that we give the maximum subscript value when declaring.

When we declare the array, we need to supply the two maxima, and when we want to access one element, we must supply 2 subscripts.

The above array ( assuming that it is to hold integers ) could be declared by:


dim a(2, 3) as integer      '( row max , col max ) 

and we could put a value in the bottom right element by:


a(2,3) = 2001

Often, we may want to perform the same operation on each element of the array, and a nested 'for' pattern can be used, e.g:

  dim row, col as integer

  for row = 0 to 2


 
for col = 0 to 3



   
a(row,col) = 0


 
next 
  next
Here, we have put 0 in each element.  The order of accessing elements in this task  is not vital, and we chose to work across a row, then move down to the next row. However this way is useful, as it matches the way the way files are input - across a row rather than down a column.   The integer variables row and col  (your choice of name ) thus take the succession of values


row
col


0

0


0

1


0

2


0

3


1

0


1

1


1

2


etc, to


2

3

Here is another example, showing how an array could be used in a noughts and crosses game program.  A complete program would be rather long, with input validation and win detection, but we will look at a few fragments of code.  Because the board is likely to be used by many functions (and because there is only one board) we might choose to make the array global, rather than pass it as an argument (shown later).

We could represent the blank board by a 3 by 3 array of char, initialised to dots, e.g:


. . .


. . .


. . .

We might declare


dim board(2,2)
as string

'
0  1  2  

and clear the board by our familiar nested 'for' pattern:


for row = 0 to 2



  for col = 0 to 2



    
board(row,col) = "."

     next 
   next 
We will need to ask the player to specify where their next move is to be - the program will know whether it is a O or X.  We will omit a check for the range 0 to 2, but will check an attempt to go in a square that is not a '.'.


'   ...get row, col number from user somehow..., then


if board(row,col)<> "."  then


 
MessageBox.show( "Cheat! - Used already!")

else


 
board(row,col) = "X"      '... etc

9. The array 'on-screen'
We could display the board in a text box by:


for row = 0 to 2



  for col = 0 to 2     'work across the cols in 1 row


    
TextBox1.AppendText(board(row,col))
     next 
     TextBox1.AppendText(VbCrLf)    ' end of a row
   next 

Users expect a better display  - not just a few X and O characters.  We could display the board using graphics commands (covered elsewhere).  We could scale-up the subscripts, so that instead of using  0,1,2,  we use 0, 100, 200.  Instead of displaying the character, we could draw an image, as in

      scaleFactor = 100


for row = 0 to 2



  for col = 0 to 2



    
if board(row,col) = "." then

             ' display a white rectangle at x position col*scaleFactor

             ' and y position = row * scaleFactor

         elseif board(row,col) = "O" then

             ' display a red circle shape at x position col*scaleFactor

             ' and y position = row * scaleFactor

         endif

     next 
   next 

Key Points

SYMBOL 183 \f "Symbol" \s 10 \h
Array elements exist in RAM and can be manipulated just like other variables.

SYMBOL 183 \f "Symbol" \s 10 \h
An individual element is selected by a subscript - an expression in ( ).  Recall that expressions include numbers, variables, and calculations.

SYMBOL 183 \f "Symbol" \s 10 \h
To perform the same operation on each element of an array, we often use a 'for' to step through the elements.

Problems - 1-dim arrays  (NB code at end of chapter)
1.
These problems are based on 12 integers - the rainfall data for each month of a year.  You will need to invent suitable numbers, and read them into an array at the start of your program.  An initial decision is whether to use:





int rainfall(12)


a).
Write a program which reads the data into an array,

      b)  then finds the largest  rainfall figure  in the array.


c).
Extend b).  so that it displays  the month number as well.


d).
Extend  c). so that it finds the average rainfall per month, by adding up the elements of the array
Problems - 2 dim arrays

1.
Write a program which sets the array    a(10, 10) to the pattern:



1 0 0 0 ...etc



0 1 0 0



0 0 1 0



.



etc


Display the array to check the result.

'-----------------------------------------------------------------------------------------------
'complete array code from earlier examples
Dim a(9) As Integer         ' 0 1 2 ...9

Dim board(2, 2) As String     ' 0  1  2... (2 dim)

'------------------------------------------------------------

Private Sub ButtonAddUp_Click(etc...)

   Dim n As Integer

   Dim sum As Integer

    sum = 0

    For n = 1 To 9

        sum = sum + a(n)

    Next

   MessageBox.show( "sum is " & sum)
End Sub

'-----------------------------------------------------------

Private Sub ButtonDisplay_Click(etc...)

    Text2.Text = ""      ' clear text box

    Dim place As Integer

    For place = 0 To 9   ' show on screen and textbox

        'join stuff onto text2.text    '(VBCrLf is newline character)

        Text2.AppendText(place & "  " & a(place) & vbCrLf)
    Next

End Sub

'-----------------------------------------------------------

Private Sub ButtonEnter_Click(etc...)

Dim place As Integer

    For place = 0 To 9

        a(place) = CInt(InputBox("Enter item " & place))
    Next

End Sub

'----------------------------------------------------------

Private Sub ButtonLargest_Click(etc...)

Dim largePlace As Integer

Dim place As Integer

largePlace = 1      ' assume element 1 is largest

For place = 2 To 9

    If a(place) > a(largePlace) Then

            largePlace = place

    End If

Next

MessageBox.show( ("largest:" & a(largePlace) & ", position:" & _

              largePlace))
End Sub

'---------------------------------------------------------

'----2 - dim code:    ------------------------------------

Private Sub ButtonSet_Click(etc...)

Dim row As Integer

Dim col As Integer

row = txtRow.Text

col = txtCol.Text

board(row, col) = txtSet.Text

End Sub

'---------------------------------------------------------

Private Sub ButtonDisplay2dim_Click(etc...)

    Dim row As Integer

    Dim col As Integer

    Text1.Text = ""  ' clear it

    For row = 0 To 2

       For col = 0 To 2  ' go across 1 line

          Text1.AppendText(board(row, col))
       Next

       Text1.Text = Text1.Text & vbCrLf  ' to start of next line

   Next

End Sub

'--------------------------------------------------------

Private Sub ButtonInit2dim_Click(etc...)

Dim row As Integer

Dim col As Integer

'clear the board: nested 'for' pattern:

    For row = 0 To 2

       For col = 0 To 2

          board(row, col) = "."

       Next

   Next

End Sub

'--------------------------------------------------------

